package org.deft.tween {
	/**
	 *  
	 * 缓动函数
	 * 
	 */	
	public class DEasing {
		/**
		 * @param t Specifies the current time, between 0 and duration inclusive.
		 * @param b 起始值
		 * @param c 结束值
		 * @param d 缓动持续时间.
		 */
		public static function easeInBack(t:Number, b:Number, c:Number, d:Number, s:Number = 0):Number {
			if (!s)
				s = 1.70158;

			return c * (t /= d) * t * ((s + 1) * t - s) + b;
		}

		public static function easeOutBack(t:Number, b:Number, c:Number, d:Number, s:Number = 0):Number {
			if (!s)
				s = 1.70158;

			return c * ((t = t / d - 1) * t * ((s + 1) * t + s) + 1) + b;
		}

		public static function easeInOutBack(t:Number, b:Number, c:Number, d:Number, s:Number = 0):Number {
			if (!s)
				s = 1.70158;

			if ((t /= d / 2) < 1)
				return c / 2 * (t * t * (((s *= (1.525)) + 1) * t - s)) + b;

			return c / 2 * ((t -= 2) * t * (((s *= (1.525)) + 1) * t + s) + 2) + b;
		}

		public static function easeOutBounce(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d) < (1 / 2.75))
				return c * (7.5625 * t * t) + b;

			else if (t < (2 / 2.75))
				return c * (7.5625 * (t -= (1.5 / 2.75)) * t + 0.75) + b;

			else if (t < (2.5 / 2.75))
				return c * (7.5625 * (t -= (2.25 / 2.75)) * t + 0.9375) + b;

			else
				return c * (7.5625 * (t -= (2.625 / 2.75)) * t + 0.984375) + b;
		}

		public static function easeInBounce(t:Number, b:Number, c:Number, d:Number):Number {
			return c - easeOutBounce(d - t, 0, c, d) + b;
		}

		public static function easeInOutBounce(t:Number, b:Number, c:Number, d:Number):Number {
			if (t < d / 2)
				return easeInBounce(t * 2, 0, c, d) * 0.5 + b;
			else
				return easeOutBounce(t * 2 - d, 0, c, d) * 0.5 + c * 0.5 + b;
		}

		public static function easeInCircular(t:Number, b:Number, c:Number, d:Number):Number {
			return -c * (Math.sqrt(1 - (t /= d) * t) - 1) + b;
		}

		public static function easeOutCircular(t:Number, b:Number, c:Number, d:Number):Number {
			return c * Math.sqrt(1 - (t = t / d - 1) * t) + b;
		}

		public static function easeInOutCircular(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d / 2) < 1)
				return -c / 2 * (Math.sqrt(1 - t * t) - 1) + b;

			return c / 2 * (Math.sqrt(1 - (t -= 2) * t) + 1) + b;
		}

		public static function easeInCubic(t:Number, b:Number, c:Number, d:Number):Number {
			return c * (t /= d) * t * t + b;
		}

		public static function easeOutCubic(t:Number, b:Number, c:Number, d:Number):Number {
			return c * ((t = t / d - 1) * t * t + 1) + b;
		}

		public static function easeInOutCubic(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d / 2) < 1)
				return c / 2 * t * t * t + b;

			return c / 2 * ((t -= 2) * t * t + 2) + b;
		}

		public static function easeInElastic(t:Number, b:Number, c:Number, d:Number, a:Number = 0, p:Number = 0):Number {
			if (t == 0)
				return b;

			if ((t /= d) == 1)
				return b + c;

			if (!p)
				p = d * 0.3;

			var s:Number;
			if (!a || a < Math.abs(c)) {
				a = c;
				s = p / 4;
			} else {
				s = p / (2 * Math.PI) * Math.asin(c / a);
			}

			return -(a * Math.pow(2, 10 * (t -= 1)) * Math.sin((t * d - s) * (2 * Math.PI) / p)) + b;
		}

		public static function easeOutElastic(t:Number, b:Number, c:Number, d:Number, a:Number = 0, p:Number = 0):Number {
			if (t == 0)
				return b;

			if ((t /= d) == 1)
				return b + c;

			if (!p)
				p = d * 0.3;

			var s:Number;
			if (!a || a < Math.abs(c)) {
				a = c;
				s = p / 4;
			} else {
				s = p / (2 * Math.PI) * Math.asin(c / a);
			}

			return a * Math.pow(2, -10 * t) * Math.sin((t * d - s) * (2 * Math.PI) / p) + c + b;
		}

		public static function easeInOutElastic(t:Number, b:Number, c:Number, d:Number, a:Number = 0,
												p:Number = 0):Number {
			if (t == 0)
				return b;

			if ((t /= d / 2) == 2)
				return b + c;

			if (!p)
				p = d * (0.3 * 1.5);

			var s:Number;
			if (!a || a < Math.abs(c)) {
				a = c;
				s = p / 4;
			} else {
				s = p / (2 * Math.PI) * Math.asin(c / a);
			}

			if (t < 1) {
				return -0.5 * (a * Math.pow(2, 10 * (t -= 1)) * Math.sin((t * d - s) * (2 * Math.PI) / p)) + b;
			}

			return a * Math.pow(2, -10 * (t -= 1)) * Math.sin((t * d - s) * (2 * Math.PI) / p) * 0.5 + c + b;
		}

		public static function easeInExponential(t:Number, b:Number, c:Number, d:Number):Number {
			return t == 0 ? b : c * Math.pow(2, 10 * (t / d - 1)) + b;
		}

		public static function easeOutExponential(t:Number, b:Number, c:Number, d:Number):Number {
			return t == d ? b + c : c * (-Math.pow(2, -10 * t / d) + 1) + b;
		}

		public static function easeInOutExponential(t:Number, b:Number, c:Number, d:Number):Number {
			if (t == 0)
				return b;

			if (t == d)
				return b + c;

			if ((t /= d / 2) < 1)
				return c / 2 * Math.pow(2, 10 * (t - 1)) + b;

			return c / 2 * (-Math.pow(2, -10 * --t) + 2) + b;
		}

		public static function easeNone(t:Number, b:Number, c:Number, d:Number):Number {
			return c * t / d + b;
		}

		public static function easeInQuadratic(t:Number, b:Number, c:Number, d:Number):Number {
			return c * (t /= d) * t + b;
		}

		public static function easeOutQuadratic(t:Number, b:Number, c:Number, d:Number):Number {
			return -c * (t /= d) * (t - 2) + b;
		}

		public static function easeInOutQuadratic(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d / 2) < 1)
				return c / 2 * t * t + b;

			return -c / 2 * ((--t) * (t - 2) - 1) + b;
		}

		public static function easeInQuartic(t:Number, b:Number, c:Number, d:Number):Number {
			return c * (t /= d) * t * t * t + b;
		}

		public static function easeOutQuartic(t:Number, b:Number, c:Number, d:Number):Number {
			return -c * ((t = t / d - 1) * t * t * t - 1) + b;
		}

		public static function easeInOutQuartic(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d / 2) < 1)
				return c / 2 * t * t * t * t + b;

			return -c / 2 * ((t -= 2) * t * t * t - 2) + b;
		}

		public static function easeInQuintic(t:Number, b:Number, c:Number, d:Number):Number {
			return c * (t /= d) * t * t * t * t + b;
		}

		public static function easeOutQuintic(t:Number, b:Number, c:Number, d:Number):Number {
			return c * ((t = t / d - 1) * t * t * t * t + 1) + b;
		}

		public static function easeInOutQuintic(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d / 2) < 1)
				return c / 2 * t * t * t * t * t + b;

			return c / 2 * ((t -= 2) * t * t * t * t + 2) + b;
		}

		public static function easeInRegular(t:Number, b:Number, c:Number, d:Number):Number {
			return c * (t /= d) * t + b;
		}

		public static function easeOutRegular(t:Number, b:Number, c:Number, d:Number):Number {
			return -c * (t /= d) * (t - 2) + b;
		}

		public static function easeInOutRegular(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d / 2) < 1)
				return c / 2 * t * t + b;

			return -c / 2 * ((--t) * (t - 2) - 1) + b;
		}

		public static function easeInSine(t:Number, b:Number, c:Number, d:Number):Number {
			return -c * Math.cos(t / d * (Math.PI / 2)) + c + b;
		}

		public static function easeOutSine(t:Number, b:Number, c:Number, d:Number):Number {
			return c * Math.sin(t / d * (Math.PI / 2)) + b;
		}

		public static function easeInOutSine(t:Number, b:Number, c:Number, d:Number):Number {
			return -c / 2 * (Math.cos(Math.PI * t / d) - 1) + b;
		}

		public static function easeInStrong(t:Number, b:Number, c:Number, d:Number):Number {
			return c * (t /= d) * t * t * t * t + b;
		}

		public static function easeOutStrong(t:Number, b:Number, c:Number, d:Number):Number {
			return c * ((t = t / d - 1) * t * t * t * t + 1) + b;
		}

		public static function easeInOutStrong(t:Number, b:Number, c:Number, d:Number):Number {
			if ((t /= d / 2) < 1)
				return c / 2 * t * t * t * t * t + b;

			return c / 2 * ((t -= 2) * t * t * t * t + 2) + b;
		}
	}
}
